MARIAN MUREŞAN

On the Maximal Orbit Transfer Problem

NB | CDF | PDF

Assume that a spacecraft is in a circular orbit and consider the problem of finding the largest possible circular orbit to which the spacecraft can be transferred with constant thrust during a set time, so that the variable parameter is the thrust-direction angle β. Also assume that there is only one center of attraction at the common center of the two circular orbits. Finally, assume normalized values for all constants and variables.

This article is divided into five sections: the orbit transfer problem, equations of motion, the optimal control problem, necessary conditions for the Mayer problem, and a dynamic approach to the maximal orbit transfer problem using *Mathematica*'s built-in `Manipulate` function.

The Earth-Mars orbit transfer problem is timely, given the successful flight and smooth landing of the American *Curiosity* rover on Mars.

The Orbit Transfer Problem

For the orbit transfer problem, assume that:

- There is a unique center of attraction.
- Initially the spacecraft moves in a circular trajectory around the center of attraction.
- The spacecraft moves with a constant thrust from a rocket engine operating in the time interval $[0, b]$.
- The spacecraft moves to the largest possible circular orbit around the center of attraction.
- The orbit transfer trajectory is coplanar with the two circular orbits and the center of attraction.

All these assumptions are stated in [1, p. 66]. Here is a sketch of a solution to the problem with some notation. The blue curve is the orbital transfer trajectory, while the red and green curves are the initial lower circular orbit and the final upper circular orbit.

```
Module[
 {f, P, Q},
 f = BezierFunction[{{3, 0}, {3.3, 1.6}, {1, 5}, {-2.3, 4.5}, {4.5 Cos[5 Pi / 6], 4.5 Sin[5 Pi / 6]}}];
 P = f[0.5];
 Q = P[[2]] (P[[2]] - 3.75) / P[[1]] + P[[1]];
 Show[
  Graphics[{
    {Red, Circle[{0, 0}, 3, {0, 3 Pi / 4}]},
    {Green, Circle[{0, 0}, 4.5, {Pi / 6, 5 Pi / 6}]},
    Arrowheads[.023],
    Arrow[{P + 0.4 {Cos[ Pi / 4 - 0.13], Sin[ Pi / 4 - 0.13]},
      P + 0.4 {Cos[ Pi / 4 - 0.135], Sin[ Pi / 4 - 0.135]}}],
    Arrow[{0.55 {Cos[ Pi / 3 + 0.31], Sin[ Pi / 3 + 0.31]},
      0.55 {Cos[ Pi / 3 + 0.32], Sin[ Pi / 3 + 0.32]}}],
    Arrow[{{0, 0}, {3, 0}}],
    Arrow[{{0, 0 }, P}],
    Arrow[{{0, 0}, 4.5 {Cos[5 Pi / 6], Sin[5 Pi / 6]}}],
    Arrow[{{P[[2]] (P[[2]] - 3.538) / P[[1]] + P[[1]], 3.538}, {Q, 3.75}}],
    Arrow[{{Q, 3.75}, {Q, 3.75} + 0.12 P}],
    Arrow[{P, {Q, 3.75} + 0.12 P}],
    {Red, Arrow[{P, P + {0.8, .65}}]},
    PointSize[0.012], Point[{3, 0}],
    Point[{4.5 Cos[5 Pi / 6], 4.5 Sin[5 Pi / 6]}],
    Point[P],
    Circle[P, 0.4, {Pi - 0.15, Pi / 4 - .14}],
    Circle[{0, 0}, 0.55, {0, Pi / 3 + 0.32}],
    Text[Style[Row[{Style["r", Italic], "(0)"}], 12], {3.45, 0.0}],
    Text[Style[Row[{Style["r", Italic], "(", Style["b", Italic], ")"}], 12],
     4.85 { Cos[5 Pi / 6], Sin[5 Pi / 6]}],
    Text[Style["v", Italic, 12], P + {-0.65, -.12}],
    Text[Style["u", Italic, 12], P + {-1.4, 0.49}],
    Text[Style["T", Italic, Red, 12], P + {0.55, 0.63}],
    Text[Style["β", 12], P + {-0.1, 0.55}],
    Text[Style["θ", 12], f[0.25] / 4 ],
    Text[Style["center of attraction", Italic, 12], {0.0, -0.3}],
    Text[Style["spacecraft", Italic, 12], P + {.85, 0.1}],
    Text[Style[Row[{Style["r", Italic], "(", Style["t", Italic], ")"}], 12], P / 2 + {0.4, 0}],
    Text[Style["initial\norbit", Italic, 12], {2.1, 1}],
    Text[Style["final\norbit", Italic, 12], {3.3, 2.3}],
    {Red, Disk[{0, 0}, 0.06]}
   }],
  ParametricPlot[f[x], {x, 0, 1}, PlotStyle -> {Blue, Thickness → 0.005}], ImageSize → {500, 300}
 ]
]
```



The notation from [1, pp. 66–68], [2], or [3] is:

- t is time in the given interval $[0, b]$, which is called the horizon.
- $r(t)$ is the radial distance from the center of attraction to the spacecraft; $r(t)$ increases as fuel is burned; $r(0) = r_0$ is the initial distance; $r(b)$ is the final and maximal distance.
- $\theta(t)$ is the polar angle, measured counterclockwise from the straight line connecting the center of attraction with the position of the spacecraft at $t = 0$.
- $u(t)$ is the radial velocity component.
- $v(t)$ is the tangential velocity component.
- $\beta(t)$ is the thrust-direction angle; it is the control variable.
- m_0 is the initial mass of the spacecraft with propellant included; $m_0 - m' t$ is the time-dependent mass, which decreases due to the constant fuel consumption rate $m' > 0$.
- T is the thrust, also assumed to be constant.
- μ is the gravitational constant.

Equations of Motion

The equations of motion of the spacecraft consistent with the above assumptions, according to [1, p. 67] and [2], are

$$r'(t) = u(t), \tag{1}$$

$$u'(t) = \frac{v^2(t)}{r(t)} - \frac{\mu}{r^2(t)} + \frac{T \sin \beta(t)}{m_0 - m' t}, \tag{2}$$

$$v'(t) = -\frac{u(t)\, v(t)}{r(t)} + \frac{T \cos \beta(t)}{m_0 - m' t}, \tag{3}$$

$$\theta'(t) = \frac{v(t)}{r(t)}. \tag{4}$$

The associated boundary conditions are

$$r(0) = r_0\,, \tag{5}$$

$$u(0) = 0, \tag{6}$$

$$v(0) = \sqrt{\frac{\mu}{r_0}}\,, \tag{7}$$

$$\theta(0) = 0, \tag{8}$$

$$u(b) = 0, \tag{9}$$

$$v(b) = \sqrt{\frac{\mu}{r(b)}}\,. \tag{10}$$

The system of nonlinear differential equations (1) to (4) with the boundary value conditions (5) to (10), the control function β, and the maximizing condition

$$\max r(b) \tag{11}$$

form the optimal control problem to be solved, assuming that the state functions r, u, v, and θ and the control function β are sufficiently smooth. Conditions (6), (7), (9), and (10) guarantee that the trajectory of the spacecraft is tangent to the two circular orbits.

The Optimal Control Problem

The goal is to maximize $r(b)$, the radius of the orbit transfer at the endpoint in time, so the cost functional is determined by

$$\Lambda[r, u, v, \theta, \beta] = r(b). \tag{12}$$

Thus the horizon is $[0, b]$ with $b > 0$. This is a Mayer optimal control problem (see Ch. 4 in [4]).

Since the differential equations (1) to (4) with conditions (5) to (10) and the cost functional (12) are not time dependent, the optimal control problem is equivalent to either of the following two problems:

- differential equations (1) to (4) with conditions (5) to (10), a given $r(b)$, and b finite and arbitrary, with optimality condition to minimize b
- differential equations (1) to (4) with conditions (5) to (10), a given $r(b)$, with the optimality condition to minimize the fuel consumption $\int_0^b (m_0 - m' t)\, dt$

Theorem 1

Under the hypotheses of Filippov's theorem (theorem 9.2.i of [4]), the optimal control problem (1) to (4) with conditions (5) to (10) and the maximizing functional (11) and (12) has an absolute maximum in the nonempty set Ω of admissible pairs.

Necessary Conditions for a Mayer Problem

For brevity, here is an abbreviated version of theorem 4.2.i in [4]: Let the Mayer problem be expressed as

$$\Lambda[x, u] = g\,(a, x\,(a), b, x(b))\ (\text{cost functional}), \tag{13}$$

$$\frac{dx}{dt} = f(t, x(t), u(t)),\ t \in [a, b],\ \text{almost everywhere } (a.e.)\ (\text{differential constraint}), \tag{14}$$

$$e[x] = (a, x(a), b, x(b)) \in B \subset \mathbb{R}^{1+n+1+n}\ (\text{boundary conditions}),$$

$$(t, x(t)) \in A,\ t \in [a, b]\ (\text{time and state constraint}),$$

$$u(t) \in U(t),\ t \in [a, b]\ (\text{control constraint}).$$

A pair $(x(t), u(t))$, $a \le t \le b$, is said to be *admissible* (or *feasible)* provided that $x : [a, b] \to \mathbb{R}^n$ is absolutely continuous [5], $u : [a, b] \to \mathbb{R}^p$ is measurable, and x and u satisfy (14) a.e. Let Ω be the class of admissible pairs (x, u). The goal is to find the minimum of the cost functional (13) over Ω, that is, to find an element $(x^*, u^*) \in \Omega$ so that $-\infty < \Lambda[x^*, u^*] \le \Lambda[x, u]$ for all $(x, u) \in \Omega$. Introduce the variables $\lambda = (\lambda_1, \ldots, \lambda_n)$, called *multipliers*, and an auxiliary function $H(t, x, u, \lambda)$, called the *Hamiltonian*, defined on $T \times U \times \mathbb{R}^n$ by

$$H(t, x, u, \lambda) = \sum_{i=1}^{n} \lambda_i\, f_i(t, x, u). \tag{15}$$

Define

$$M(t, x, \lambda) = \inf_{u \in U(t)} H(t, x, u, \lambda).$$

Further necessary assumptions:

1. There exists an element $(x^*, u^*) \in \Omega$ such that $-\infty < \Lambda[x^*, u^*] \le \Lambda[x, u]$ for all $(x, u) \in \Omega$.
2. A is closed in $\mathbb{R}^{1+n}$.
3. The set $S = \{(t, x, u) \mid (t, x) \in A, u \in U(t)\}$ is closed in $\mathbb{R}^{1+n+p}$.
4. $f \in C^1(S, \mathbb{R})$.
5. Notation:

   $f_{it} = \frac{\partial f_i}{\partial t}$, $f_{ix_j} = \frac{\partial f_i}{\partial x_j}$, $H_{x_j} = \frac{\partial H}{\partial x_j} = \sum_{i=1}^{n} \lambda_i\, f_{ix_j}$, $H_t = \frac{\partial H}{\partial t} = \sum_{i=1}^{n} \lambda_i\, f_{it}$, $H_{\lambda_j} = \frac{\partial H}{\partial \lambda_j} = f_j$.

6. The graph $\{(t, x^*(t)) \mid a \le t \le b\}$ of the optimal trajectory x^* belongs to the interior of A.
7. U does not depend on time and is a closed set.

8. The endpoint $e(x^*) = (a, x^*(a), b, x^*(b))$ of the optimal trajectory x^* is a point of B, where B has a tangent variety B' (of some dimension δ, $0 \le \delta \le 2n+2$) whose vectors are denoted by

$$h = (a, \xi_1, b, \xi_2), \xi_1 = \left(\xi_1^1, \ldots, \xi_1^n\right), \xi_2 = \left(\xi_2^1, \ldots, \xi_2^n\right),$$

or by

$$h = (da, dx_1, db, dx_2), dx_1 = \left(d\xi_1^1, \ldots, d\xi_1^n\right), dx_2 = \left(d\xi_2^1, \ldots, d\xi_2^n\right).$$

Theorem 2

Assume the above eight hypotheses and let (x^*, u^*) be an optimal pair for the Mayer problem (13) and (14). Then the optimal pair (x^*, u^*) necessarily has the following properties:

(a) There exists an absolutely continuous function $\lambda(t) = (\lambda_1(t), \ldots, \lambda_n(t))$ such that

$$\frac{d\lambda_i}{dt} = -H_{x_i}(t, x^*(t), u^*(t), \lambda(t)), i = 1, \ldots, n, t \in [a, b]\,(a.e.).$$

If dg is not identically zero at $e[x^*]$, then $\lambda(t)$ is never zero in $[a, b]$.

(b) For almost any fixed $t \in [a, b]$ (a.e.), the Hamiltonian, as a function depending only on u, takes its minimum value in U at the optimal strategy $u^* = u^*(t)$. This implies $M(t, x^*(t), \lambda(t)) = H(t, x^*(t), u^*(t), \lambda(t))$, $t \in [a, b]$ (a.e).

(c) The function $M(t) = M(t, x^*(t), \lambda(t))$ coincides a.e. in $[a, b]$ with an absolutely continuous function, and

$$\frac{dM}{dt} = \frac{d}{dt} M(t, x^*(t), \lambda(t)) = H_t(t, x^*(t), u^*(t), \lambda(t)), t \in [a, b]\,(a.e.).$$

(d) (transversality relation) There exists a constant $\lambda_0 \geqslant 0$ such that

$$(\lambda_0 g_a - M(a))\, da + \sum_{i=1}^{n}\left(\lambda_0 g_{x_1^i} + \lambda_i(a)\right) dx_1^i + (\lambda_0 g_b + M(b))\, db + \sum_{i=1}^{n}\left(\lambda_0 g_{x_2^i} - \lambda_i(b)\right) dx_i^2 = 0, \tag{16}$$

for every vector $h = (da, dx_1, db, dx_2) \in B'$.

From (15) and (a) of theorem 2, the Hamiltonian and the equations for the multipliers for (1) to (4) are

$$H = \lambda_1 u(t) + \lambda_2 \left(\frac{v^2(t)}{r(t)} - \frac{\mu}{r^2(t)} + \frac{T \sin\beta(t)}{m_0 - m' t}\right) + \lambda_3\left(-\frac{u(t)\,v(t)}{r(t)} + \frac{T\cos\beta(t)}{m_0 - m' t}\right) + \lambda_4 \frac{v(t)}{r(t)}, \tag{17}$$

$$\lambda_1'(t) = -\frac{\partial H}{\partial r} = -\lambda_2(t)\left(-\frac{v^2(t)}{r^2(t)} + \frac{2\mu}{r^3(t)}\right) - \lambda_3(t)\left(\frac{u(t)\,v(t)}{r^2(t)}\right) - \lambda_4(t)\left(-\frac{v(t)}{r^2(t)}\right), \tag{18}$$

$$\lambda_2'(t) = -\frac{\partial H}{\partial u} = -\lambda_1(t) + \lambda_3(t)\frac{v(t)}{r(t)}, \tag{19}$$

$$\lambda_3'(t) = -\frac{\partial H}{\partial v} = -\lambda_2(t)\frac{2\,v(t)}{r(t)} + \lambda_3(t)\frac{u(t)}{r(t)} - \lambda_4(t)\frac{1}{r(t)}, \tag{20}$$

$$\lambda_4'(t) = 0. \tag{21}$$

From (21) and (18), $\lambda_4 \equiv 0$ and thus (17) to (20) become

$$H = \lambda_1 u(t) + \lambda_2\left(\frac{v^2(t)}{r(t)} - \frac{\mu}{r^2(t)} + \frac{T\sin\beta(t)}{m_0 - m' t}\right) + \lambda_3\left(-\frac{u(t)\,v(t)}{r(t)} + \frac{T\cos\beta(t)}{m_0 - m' t}\right),$$

$$\lambda_1'(t) = -\lambda_2(t)\left(-\frac{v^2(t)}{r^2(t)} + \frac{2\mu}{r^3(t)}\right) - \lambda_3(t)\left(\frac{u(t)\,v(t)}{r^2(t)}\right), \tag{22}$$

$$\lambda_2'(t) = -\lambda_1(t) + \lambda_3(t)\frac{v(t)}{r(t)}, \tag{23}$$

$$\lambda_3'(t) = -\lambda_2(t)\frac{2\,v(t)}{r(t)} + \lambda_3(t)\frac{u(t)}{r(t)}. \tag{24}$$

Furthermore, from (b) in theorem 2,

$$0 = \frac{\partial H}{\partial \beta} = \frac{T}{m_0 - m' t}(\lambda_2(t)\cos\beta(t) - \lambda_1(t)\sin\beta(t)) \Longrightarrow \beta(t) = \arctan\frac{\lambda_2(t)}{\lambda_3(t)}.$$

Thus the control function β is determined by the multipliers λ_2 and λ_3.

Based on (4), note that the polar angle θ is determined by v and r.

From the transversality relation (d) in theorem 2 (i.e. equation (16)),

$$\lambda_1(b) = 1. \tag{25}$$

This yields a system of six nonlinear differential equations (1), (2), (3), (22), (23), and (24) in the variables r, u, v, λ_1, λ_2, and λ_3 with six bilocal conditions (5), (6), (7), (9), (10), and (25).

As mentioned earlier, the variables β and θ follow.

The next section implements a dynamical approach to the maximal orbit transfer problem.

A Dynamic Approach to the Maximal Orbit Transfer Problem

The function `MaximalRadiusOrbitTransfer` dynamically shows the maximal radius orbit transfer between two coplanar circular orbits so that their centers are located at a single center of attraction. Here `thrust` is the constant thrust of the engine, `dmr` is the decreasing mass rate due to the constant propellant flow rate, `b` is the final time, `m0` is the initial mass of the spacecraft including the propellant, `μ` is the gravitational constant, `r0` is the initial radius, `u0` is the initial radial velocity, `ub` is the final radial velocity, `v0` is the initial tangential velocity, and `k` is the number of thrust vectors.

Clearly the problem is nonlinear and, to the author’s knowledge, no closed-form solution has been found. The possibility of obtaining a solution through a numerical method remains, as implied by theorem 1. The accuracy of the results depends sensitively on the initial values. The `Method` option is needed for *Mathematica* 9 or lower; for faster processing, remove it in *Mathematica* 10 or higher.

```
MaximalRadiusOrbitTransfer[thrust_, dmr_, b_, r0_, μ_, m0_, u0_, ub_, k_] :=
 Module[{v0, t, r, u, v, λr, λv, λu, sol, R, V, θ, θmax},
  (* data for the flight *)
  v0 = Sqrt[μ / r0];    (* initial tangential velocity; tangent to the inner circle *)
  sol = NDSolve[{      (* solution of the problem *)
      r'[t] == u[t],
      u'[t] == v[t]^2 / r[t] - μ / r[t]^2 + thrust / (m0 - dmr t) λu[t] / Sqrt[λu[t]^2 + λv[t]^2],
      v'[t] == -((u[t] v[t]) / r[t]) + thrust / (m0 - dmr t) λv[t] / Sqrt[λu[t]^2 + λv[t]^2],
      λr'[t] == λu[t] (v[t]^2 / r[t]^2 - 2.0 μ / r[t]^3) - λv[t] (u[t] v[t]) / r[t]^2,
      λu'[t] == -λr[t] + λv[t] v[t] / r[t],
      λv'[t] == -2.0 λu[t] v[t] / r[t] + λv[t] u[t] / r[t],
      r[0] == r0,
      u[0] == u0,
      v[0] == v0,
      λr[b] == 1.0,
      u[b] == ub,
      v[b] == Sqrt[μ / r[b]]},
    {r, u, v, λr, λu, λv}, {t, 0, b},
    Method → {"Shooting", "StartingInitialConditions" →
       {r[0] == r0, u[0] == u0, v[0] == v0, λr[0] == 1.0, λu[0] == -0.5, λv[0] == 0.0}}];
  R[t_] = Evaluate[r[t] /. sol][[1]];
  V[t_] = Evaluate[v[t] /. sol][[1]];
  (* the orbit transfer figure *)
  θ[time_] :=
   NIntegrate[Interpolation[Table[{n b / k, V[n b / k] / R[n b / k]}, {n, 0, k}],
      InterpolationOrder -> 1][t], {t, 0, time}];
  θmax = 180 θ[b] / N[Pi];
  Show[
   Graphics[{
      Text[Style[Row[{"final radius = ", R[b]}], 12], {0, .36 r0}],
      Text[Style[Row[{"final θ = ", Superscript[θmax, "∘"]}], 12], {0, .19 r0}],
      {EdgeForm[Thin], Red, Disk[{0, 0}, 0.04]},
      {Red, Circle[{0, 0}, r0, {0, 2 Pi}]},
      {Green, Circle[{0, 0}, R[b], {0, 2 Pi}]},
      {Red, Arrowheads[0.02],
       Arrow[Table[{R[n b / k] {Cos[θ[n b / k]], Sin[θ[n b / k]]},
          R[n b / k] {Cos[θ[n b / k]], Sin[θ[n b / k]]} + thrust {Sin[θ[n b / k]], Cos[θ[n b / k]]}},
         {n, 0, k}]]}
     }],
   ListLinePlot[Table[R[n b / k] {Cos[θ[n b / k]], Sin[θ[n b / k]]}, {n, 0, k}],
    PlotStyle → {Blue, Thick}],
   Axes → False, PlotRange → All, ImageSize → {400, 400}, AxesOrigin → {0, 0}]
 ]

Manipulate[
 Quiet@MaximalRadiusOrbitTransfer[thrust, dmr, b, 1.0, 1.0, 1.0, 0.0, 0.0, k],
 {{thrust, 0.1405}, 0.14, 0.15, 0.0001, Appearance → "Labeled"},
 {{dmr, 0.07485, "decreasing mass rate"}, 0.07, 0.08, 0.001, Appearance → "Labeled"},
 {{b, 3.317291, "final time interval"}, 3.2, 3.5, 0.01, Appearance → "Labeled"},
 {{k, 20, "number of thrust arrows"}, 15, 50, 1, Appearance → "Labeled"},
 SaveDefinitions → True, TrackedSymbols → {thrust, dmr, b, k}, SynchronousUpdating → False
]
```



A similar picture can be found on the front cover and on pages 1–2 of [6].

Acknowledgments

The author expresses his deep gratitude to Prof. Dr., Dr. H. C., Heiner H. Gonska of the University of Duisburg-Essen, Faculty of Mathematics, located in Duisburg, Germany, for his kind invitation and warm hospitality. The invitation was funded by a grant from the Center of Excellence for Applications of Mathematics supported by DAAD. The author also thanks Aida Viziru for her rapid and professional help.

References

[1] A. E. Bryson and Y.-C. Ho, *Applied Optimal Control: Optimization, Estimation, and Control*, New York: Halsted Press, 1975.

[2] S. Alfano and J. D. Thorne, *Constant-Thrust Orbit-Raising Transfer Charts*, Report PL-TR-93-1010, July 1993. www.dtic.mil/dtic/tr/fulltext/u2/a269088.pdf.

[3] S. Alfano and J. D. Thorne, "Circle-to-Circle Constant-Thrust Orbit Raising," *The Journal of the Astronautical Sciences*, **42**(1), 1994 pp. 35–45. www.centerforspace.com/downloads/files/pubs/JAS.V42.N01.pdf.

[4] L. Cesari, *Optimization—Theory and Applications, Problems with Ordinary Differential Equations*, *Applications of Mathematics*, Vol. 17, New York: Springer, 1983.

[5] M. Mureşan, *A Concrete Approach to Classical Analysis*, New York: Springer, 2009.

[6] R. Vinter, *Optimal Control*, *Systems & Control: Foundations & Applications*, Boston: Birkhäuser, 2000.

M. Mureşan, "On the Maximal Orbit Transfer Problem," *The Mathematica Journal*, 2015. dx.doi.org/doi:10.3888/tmj.17-4.

About the Author

Marian Mureşan is affiliated with Babeş-Bolyai University, Faculty of Mathematics and Computer Science, in Cluj-Napoca, Romania. He is interested in analysis, calculus of variations, optimal control, and nonsmooth analysis.

Marian Mureşan
Babeş-Bolyai University
Faculty of Mathematics and Computer Science
1, M. Kogălniceanu str., 400084, Cluj-Napoca
Romania
mmarianus24@yahoo.com
mmarian@math.ubbcluj.ro